
Folder Size Tracker 

Introducción

Cuando el espacio libre en su disco duro va desapareciendo y cada  vez lo encuentra más ocupado, la respuesta lógica es identificar las carpetas que están consumiendo más espacio.   Es simple; cuenta con el Explorador de Windows para recopilar la información.  Abra el Explorador y comience el recorrido, carpeta por carpeta, dando un clic-derecho en cada una, seleccionando propiedades y revisando el tamaño.  Sin embargo es una tarea larga y tediosa, puesto que debe hacerlo carpeta por carpeta, registrando manualmente para cada una el tamaño informado.

El FolderSizeTracker es una herramienta que ofrece una mejor alternativa al enfoque manual.   Para averiguar cuales son las carpetas que más espacio consumen, basta escribir la ruta que desea investigar en éste programa y dejarlo actuar.  El resultado es recopilado en una hoja de Excel, y desplegado en gráficos de pastel 3-D que giran, haciendo realmente fácil detectar las carpetas que ocupan más espacio.

El FolderSizeTracker funciona en todas las versiones de Windows y con Excel 2000 en adelante.

Instalación

Abra Notepad.  Copie el Script que se encuentra al final de este documento en el Notepad. Guarde el archivo en un directorio adecuado de su equipo, con el nombre FolderSizeTracker.vbs

Funcionamiento

Para utilizar FolderSizeTracker, simplemente un doble-clic en el archivo.  Tan pronto lo haga, verá una pequeña caja de diálogo preguntando por el disco o ruta del directorio con carpetas que desea investigar.  Digite la información y oprima OK.  La herramienta recopila los tamaños de las carpetas, ingresa la información en una hoja de Excel y compila la información en un gráfico pastel 3-D.  El gráfico 3-D gira 360º y se devuelve, mostrando brevemente el pastel completo.

Cuando el gráfico termina de girar, aparece una serie de cajas de diálogo con instrucciones como las que contiene la tabla siguiente.  Tiene la opción de deshabilitar el despliegue automático de instrucciones; si posteriormente desea verlas de nuevo, localice y borre el archivo FolderSizeTracker.ini

	Opciones

	Para identificar mejor las porciones pequeñas, puede rotar manualmente el gráfico:

Despliegue el menú Gráfico.

Seleccione el comando de vista 3-D.

Clic en uno de los botones de rotación.

Clic al botón Aplicar.

Para identificar la carpeta representada por un pedazo en particular, pase el mouse sobre esa tajada.   Aparece un pop-up con el nombre de la carpeta y el tamaño en MB (FolderSizeTracker no muestra carpetas con 1 MB o menos.)


Para ver la lista completa de las carpetas que han sido graficadas, con sus tamaños, de un clic a  la mirilla Carpetas  en el libro Excel.  Al cerrar Excel, éste preguntará si desea guardar el libro.

Si ninguno de los directorios en la ruta que escribió contiene más de 1 MB de datos, FolderSizeTracker  despliega un mensaje de aviso, y se cierra. 

‘Script:

'Option Explicit

Dim WshShell, FileSystem, XL, ThePath, Row, TheFile

Set WshShell = WScript.CreateObject("WScript.Shell")

Set FileSystem = Wscript.CreateObject("Scripting.FileSystemObject")

Set XL = WScript.CreateObject("Excel.Application")

'Main Routine


CurDir = WshShell.CurrentDirectory


TheFile = CurDir + "\FolderSizeTracker.ini"

  
Call GetThePath

  
Call SplashScreen  

  
Call CreateSpreadSheet


Call BuildSpreadSheet


Call SortData


Call CreateChart


Call RotateChart


If (FileSystem.FileExists(TheFile)) Then



'Do Nothing


Else



Call Instructions


End If

'End Main Routine

'******************************************************************

Sub SplashScreen

  
Splash = "Por favor aguarde mientras FolderSizeTracker analiza el contenido de la carpeta" & vbCrLf &_

  
"y diciona los datos a la hoja Excel." & vbCrLf & vbCrLf &_

  
"Created by Greg Shultz" & vbCrLf &_

  
"www.TheWinWiz.com"

  
Dummy = WshShell.Popup(Splash, 2, "Folder Size Tracker", 64)

End Sub 'SplashScreen

'******************************************************************

Sub GetThePath

  
ThePrompt = "Digite el drive o ruta a la que desea verificar tamaños de carpetas." & vbCrLf & vbCrLf &_

  
"Por ejemplo:  C:\   o   C:\Documents and Settings"

  
ThePath = InputBox(ThePrompt, "Folder Size Tracker")

  
If ThePath = "" Then WScript.Quit

  
If Right(ThePath,1) = ":" Then

  

ThePath = ThePath & "\"

  
End If

End Sub 'GetThePath

'******************************************************************

Sub CreateSpreadSheet

   
XL.Visible = True

   
XL.WorkBooks.Add

   
XL.Worksheets(2).visible = False

   
XL.Worksheets(3).visible = False

   
XL.ActiveSheet.Name = "Carpetas"

   
XL.Range("A1").Select

   
XL.Columns(1).ColumnWidth = 30

   
XL.Columns(2).ColumnWidth = 10

   
XL.Columns(2).NumberFormat = "#,##0.0"

   
XL.ActiveSheet.Cells(1,1).Value = "Directorios en: " & ThePath

   
XL.Range("A1:A1").Select

   
XL.Selection.Font.Size = 14

   
XL.Range("A2:B2").Select

   
XL.Selection.Font.Size = 12

   
XL.ActiveSheet.Cells(3,1).Value = "Directorio"

   
XL.ActiveSheet.Cells(3,2).Value = "Total (MB)"

   
XL.ActiveSheet.Cells(3,2).AddComment "FolderSizeTracker no muestra directorios con 1MB o menos."

   
XL.Range("A1:B3").Select

    XL.Selection.Font.Bold = True

   
XL.Range("A5:B5").Select

   
Row = 5

End Sub 'CreateSpreadSheet

'******************************************************************

Sub BuildSpreadSheet


Set CurrentFolder = FileSystem.GetFolder(ThePath)


For Each Folder In CurrentFolder.SubFolders



If Folder.Name <> "System Volume Information" Then




FolderSize = Folder.Size




FolderSize = (FormatNumber(FolderSize, 0, , , 0)/1024)/1024 


 

If FolderSize > 1 Then





XL.ActiveSheet.Cells(Row,1).Value = Folder.Name

           

XL.ActiveSheet.Cells(Row,2).Value = FolderSize





Row = Row + 1

      

End If

      
End If


Next

End Sub 'BuildSpreadSheet

'******************************************************************

Sub SortData


If Row = 5 Then



Dummy = MsgBox("Folder Size Tracker no encontró directorios en la ruta " & vbcrlf & vbcrlf &_ 



ThePath & vbcrlf & vbcrlf &_



"por encima de 1MB!" & vbcrlf & vbcrlf &_



"FolderSizeTracker terminará.", 16, "Folder Size Tracker") 



XL.ActiveWorkBook.Saved = True



XL.WorkBooks.Close



XL.Application.Quit



Wscript.Quit


Else



XL.ActiveCell.CurrentRegion.Select



XL.Selection.Sort XL.Worksheets(1).Range("B3"), 2, , , , , , 0, 1, False, 1


End If

End Sub 'SortData

'******************************************************************

Sub CreateChart


XL.Charts.Add


XL.ActiveChart.ChartType = 70


XL.ActiveChart.Name = "Pie Chart"




XL.ActiveChart.PlotArea.Select


XL.Selection.Left = 1


XL.Selection.Top = 1


XL.Selection.Height = 100


XL.Selection.Width = 675



XL.ActiveChart.HasLegend = True



XL.ActiveChart.Legend.Position = -4107 

End Sub 'CreateChart

'******************************************************************

Sub RotateChart


For intRotate = 10 To 360 Step 10



WScript.Sleep(100)

    
XL.ActiveChart.Rotation = intRotate


Next


For intRotate = 350 To 0 Step -10



WScript.Sleep(100)

    
XL.ActiveChart.Rotation = intRotate


Next


XL.ActiveChart.ChartArea.Select


XL.ActiveChart.Deselect

End Sub 'RotateChart

'******************************************************************

Sub Instructions


ThePrompt = "Para identificar mejor las tajadas pequeñas, puede rotar manualmente el gráfico:" & vbcrlf & vbcrlf & _


"1) Despliegue el menú Chart" & vbcrlf &_


"2) Seleccione el comando de 3-D View" & vbcrlf &_


"3) Click en cualquiera de los botones Rotate" & vbcrlf &_ 


"4) Click en el botón Aplicar"


Dummy = MsgBox(ThePrompt, 64, "Folder Size Tracker")


ThePrompt = "Para identificar la carpeta representada por una tajada en particular," & vbcrlf &_


"simplemente pase el mouse sobre esa tajada." & vbcrlf & vbcrlf & _


"El valor desplegado es el tamaño del directorio en MB."


Dummy = MsgBox(ThePrompt, 64, "Folder Size Tracker")


ThePrompt = "Recuerde que Folder Size Tracker no muestra carpetas" & vbcrlf &_ 


"conteniendo menos de 1MB."


Dummy = MsgBox(ThePrompt, 64, "Folder Size Tracker")


ThePrompt = "¿Desea poder ver éstas instrucciones de nuevo?"


Dummy = MsgBox(ThePrompt, 4, "Folder Size Tracker")


If Dummy = vbNo Then



ThePrompt = "Si desea ver estas instrucciones en otra oportunidad, localice y borre el archivo FolderSizeTracker.ini"



Dummy = MsgBox(ThePrompt, 64, "Folder Size Tracker")



Set AFile = FileSystem.CreateTextFile(TheFile, True)


End If

End Sub 'Instructions

